# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=pantheon-notifications
pkgver=6.0.3
pkgrel=1
pkgdesc='Gtk Notifications Server'
arch=(loong64 x86_64)
url=https://github.com/elementary/notifications
license=(GPL3)
groups=(pantheon)
depends=(
  glib2
  gtk3
  libcanberra.so
  libcanberra-gtk3.so
  libgranite.so
  libhandy-1.so
)
makedepends=(
  git
  intltool
  meson
  vala
)
_tag=ac4a171d98b3da6d6fea29554c932c1fc78cb40c
source=(pantheon-notifications::git+https://github.com/elementary/notifications.git#tag=${_tag})
b2sums=(SKIP)

pkgver() {
  cd pantheon-notifications
  git describe --tags
}

build() {
  arch-meson pantheon-notifications build
  ninja -C build
}

package() {
  DESTDIR="${pkgdir}" ninja -C build install
}

# vim: ts=2 sw=2 et:
